
Crystal est un langage de programmation, encore jeune. Il s'inspire de Ruby pour la syntaxe mais vise des performances proches du C. La version 0.16 vient de sortir, avec un nouvel algorithme pour l'inférence de types. À noter, le compilateur de Crystal est écrit en Crystal.


Voici à quoi ressemble un serveur HTTP basique écrit en Crystal :


# A very basic HTTP server
require "http/server"

server = HTTP::Server.new(8080) do |context|
  context.response.content_type = "text/plain"
  context.response.print "Hello world, got #{context.request.path}!"
end

puts "Listening on http://0.0.0.0:8080"
server.listen

Une syntaxe proche de Ruby


Comme on peut le voir dans l'exemple ci-dessus, la syntaxe est très proche de celle de Ruby. C'est un des objectifs du langage. Ceci dit, la compatibilité avec Ruby n'en fait pas partie.


On retrouve ainsi des classes similaires à celles de Ruby :


class Person
  def initialize(name : String)
    @name = name
    @age = 0
  end

  def name
    @name
  end

  def age
    @age
  end
end


Ou la notion de blocks :


def twice
  yield
  yield
end

twice do
  puts "Hello!"
end

Typage statique


Mais il existe également des différences avec Ruby. Par exemple, Crystal a un typage statique et des génériques :


class MyBox(T)
  def initialize(@value : T)
  end

  def value
    @value
  end
end

int_box = MyBox(Int32).new(1)
int_box.value # => 1 (Int32)

string_box = MyBox(String).new("hello")
string_box.value # => "hello" (String)


En général, l'inférence de types permet de ne pas spécifier les types des variables et méthodes, mais vous pouvez de manière optionnelle le faire :


def add(x : Number, y : Number)
  x + y
end

Bindings avec le langage C


Crystal permet de déclarer des bindings avec le langage C. Et, fait notable, ces bindings peuvent être déclarés en Crystal. Voici un exemple pris de la bibliothèque standard pour s'interfacer avec la libyaml :


@[Link("yaml")]
lib LibYAML
  alias Int = LibC::Int

  PARSER_SIZE = 480
  type Parser = Void*

  struct VersionDirective
    major : Int
    minor : Int
  end

  # ...

  fun yaml_parser_initialize(parser : Parser*) : Int
end

Génération et évaluation de code au moment de la compilation


Le meta-programming de Ruby est très puissant mais aussi très coûteux en performances. Crystal a donc choisi une autre voie : les macros. Celles-ci permettent d'éviter d'avoir à écrire beaucoup de code répétitif sans impacter de manière notables les performances.


Une macro reçoit un noeud AST et va générer du code lors de la compilation.


macro define_method(name, content)
  def {{name}}
    {{content}}
  end
end

# Va générer:
#
#     def foo
#       1
#     end
define_method foo, 1

foo #=> 1

Concurrence et parallélisme


Crystal permet actuellement de construire des programmes concurrents avec des primitives comme les fibers et channels, mais la jeunesse du langage se fait sentir sur ces aspects.


On peut séparer des tâches à effectuer dans des fibers. Chaque fiber a un contexte propre et est comparable à une goroutine en Go par exemple. Les fibers sont coopératives et ne vont passer la main à une autre fiber que de manière explicite (par opposition aux threads, qui sont pré-emptifs).


Il existe également une fiber un peu particulière, l'Event Loop, qui gère tous les I/O.


Enfin, les fibers communiquent entre elles via des channels.


require "socket"

channel = Channel(String).new

spawn do
  server = TCPServer.new("0.0.0.0", 8080)
  socket = server.accept
  while line = socket.gets
    channel.send(line)
  end
end

spawn do
  while line = gets
    channel.send(line)
  end
end

3.times do
  puts channel.receive
end


Ainsi, Crystal permet déjà de structurer son code pour gérer la concurrence, mais est limité de la même façon que le JavaScript pour le parallélisme : une seule fibre peut s'exécuter à un moment donné.

Compilation vers du code natif efficace


Crystal profite de LLVM pour compiler vers du code natif. Les performances sont au rendez-vous. Par exemple, dans ce benchmark, Crystal se retrouve entre C/C++ et Go/Node (et donc loin devant Ruby).

